

Grand Theft Cygnus

Storyline

As we all know, the swans all belong to the Queen. As the hero of our game, you are intent on stealing as many of them as you can. However, Her Majesty is keeping a lookout and will blast you with her shotgun if you get too close.

Move your noose with arrow keys, and when you're correctly lined up, press space to lasso and steal the swan. As long as it's not the one HM is hiding behind, then you'll receive a hint about how far away she is. If you see 0 or 1, run away! If you hear her gunshot, it's game over. If you successfully manage to creep around the monarch and steal all but one of the swans (the one she's hiding behind) then you'll move on to another part of the river.

To accompany my USB keyboard interface for the Minstrel 4th, and in the spirit of magazine type-ins, which were the thing in the heyday of the Jupiter Ace, I'm providing this game as a type-in.

I hope you enjoy the process of keying it in and playing the game.

- Shiela

```
0 variable totswans
0 variable swansleft
0 variable totscore
32 variable chrundercsr
0 variable lastkey
```

```
1 constant ALLGOOD
2 constant WIN
3 constant LOSE
```

```
ALLGOOD variable gamestate
```

```
create swanxs 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c,
create swanys 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c, 0 c,
```

```
9551 variable playerpos
```

```
0 variable SEED
```

```
: SEEDON
  SEED @ 75 U* 75 0 D+
  OVER OVER U< --
  1- dup seed !
;
```

```
: RND
  SEEDON U* SWAP DROP
;
```

```
: RAND
  ?DUP 0=
  IF
  15403 @ SWAP
  THEN
  SEED !
;
```

```
DEFINER CODE
```

```
DOES>
  CALL
;
```

```
CODE HALT 118 C, 253 C, 233 C,
```

```
: PAUSE ( length of pause in 50ths / sec -- )
```

```
  0
  DO
  HALT
  LOOP
;
```

```
: VINC ( a -- )
  dup @ 1+ swap !
;
```

```
: VDEC ( a -- )
  dup @ 1- swap !
;
```

```
: Y/N
  ." (Y/N)?" BEGIN
```

```

INKEY DUP ASCII y = IF
DROP 2 ELSE
ASCII n = THEN
?DUP UNTIL 1-
;

: ?CONT
CLS
." Caught in the act!" CR
." You stole " totscore @ . ." swans" CR
." Another game?" Y/N
;

: SQ
DUP *
;

: SQR ( f - sqr root )
1. 10 0
DO
4 PICK 4 PICK 4 PICK 4 PICK F/ F+ .5 F*
LOOP
4 ROLL 4 ROLL DROP DROP
;

: PYTHAG ( a b -- h )
sq swap sq +
ufloat sqrt int
;

: GR ( 8 bytes of data, ASCII code -- )
8 * 11263 + dup
8 +
do
i c! -1
+loop
;

: BM
( assets )
60 126 231 227 195 67 67 70          1 GR
14 28 56 125 255 255 255 127        2 GR
0 0 0 0 0 15 62 126                 3 GR
124 248 249 251 255 255 254 252 4   4 GR
0 0 192 128 0 0 0 0                 5 GR
170 85 170 85 170 85 170 85         6 GR
;

: scrnadd ( x y -- a )
32 *
9216 + +
;

: playerx ( -- x )
playerpos @ 9216 -
32 mod
;

: playery ( -- y )
playerpos @ 9216 -
32 /
;

```

```

: drw_swan_at ( x y -- )
  scrnadd

  1 over c!
  32 +
  2 over c!
  31 -
  3 over c!
  32 +
  4 over c!
  1 +
  5 over c!

  drop

;

: undraw_swan_at ( add -- )

  32 over c!
  1 +
  32 over c!
  31 +
  32 over c!
  1 +
  32 over c!
  1 +
  32 over c!

  drop

;

: newswanxy ( -- x y )
  begin
    10 rnd 3 *
    11 rnd 2 *

    over over scrnadd c@
    32 = if
      1
    else
      drop drop 0
    then
  until

;

: swangen
  totswans @ 0 do
    newswanxy ( -- x y )
    over over
    swanys i + c!
    swanxs i + c!
    drw_swan_at
  loop

;

: blinkcsr
  playerpos @ c@
  chrundercsr !
  6 playerpos @ c!

;

: unblinkcsr
  chrundercsr @
  playerpos @ c!

;

: bling
  119 50 beep

```

```

95 50 beep
80 100 beep
;

: uhoh
3034 400 beep 5 pause 3822 1000 beep
;

: findhead
chrundercsr @
dup 2 = if
    playerpos @ 32 - playerpos !
then
dup 3 = if
    playerpos vdec
then
dup 4 = if
    playerpos @ 33 - playerpos !
then
dup 5 = if
    playerpos @ 34 - playerpos !
then
drop
;

: calculate_distance
( is it the queen? she is behind swan zero )
swanxs c@ swanys c@
scrnadd playerpos @
= if
    LOSE gamestate !
else
( distance )
    playerx swanxs c@ - abs 3 /
    playery swanys c@ - abs 2 /
    pythag

    dup 9 > if
        drop 9
    then

    48 + playerpos @ c!

    swansleft vdec
    swansleft @ 1 = if
        WIN gamestate !
        bling bling bling
    then
then
;

: chkkey
inkey
dup 0 > if
    dup lastkey !
then
drop
;

: handlefire
chrundercsr @ 6 < if
    findhead
    playerpos @ undraw_swan_at

```

```

calculate_distance

( space means break during beep )
begin inkey 1- until

gamestate @ ALLGOOD = if
  bling
  totscore vinc
then
gamestate @ LOSE = if
  uhoh
then

then

;

: handlekey
lastkey @
dup 1 = if
  ( L )
  playerpos @ 9216 > if
    playerpos vdec
  then

then
dup 3 = if
  ( R )
  playerpos @ 9920 < if
    playerpos vinc
  then

then
dup 7 = if
  ( U )
  playerpos @ 9248 > if
    playerpos @ 32 - playerpos !
  then

then
dup 9 = if
  ( D )
  playerpos @ 9888 < if
    playerpos @ 32 + playerpos !
  then

then

dup 32 = if
  ( fire )
  handlefire

then

drop
0 lastkey !

;

: initgame
BM
CLS
12 totswans !
0 totscore !
15403 c@ seed !

;

```

```

: initlevel ( n -- )
  swansleft !
  ALLGOOD gamestate !
  9551 playerpos !
  CLS
  swangen
;

: GTC
  initgame
  12 initlevel

  begin
    begin
      blinkcsr
      chkkey
      3 pause

      unblinkcsr
      handlekey
      chkkey
      3 pause

      gamestate @ ALLGOOD = 1-
    until

    gamestate @ WIN = if
      ( win level )
      totswans vdec
      totswans @ initlevel
      0
    else ( lose )
      ?cont 1-
      dup 0= if
        12 totswans !
        12 initlevel
        0 totscore !
      then
    then

  until
;

```